
ABSTRACT 

Correcting correlation coefficients for attenuation resulting from measurement error has a long history (dating back to 
Spearman’s work in 1904). However, the estimation of confidence intervals for disattenuated correlation coefficients is 
a more recent concern. The macro described in this paper computes the point estimate of the correlation coefficient 
after adjusting for measurement error, and the endpoints of two estimates of confidence intervals: the traditional 
Fisher/Spearman interval and the confidence set suggested by Charles (2005). The macro is written in Base SAS 
software, with arguments consisting of the obtained sample correlation coefficient, estimates of the reliability of each 
variable, sample size, and level of confidence for the intervals. Through providing a convenient method to compute 
disattenuated correlation coefficients and accurate confidence intervals, this paper will encourage researchers to be 
mindful of the impact of measurement error on indices of correlation, as well as the consideration of sampling error in 
the interpretation of results. 

INTRODUCTION 

Indices of correlation are used to describe the direction and strength of the relationship between two variables. The 
most frequently used index of correlation is the Pearson Product-Moment Correlation Coefficient: 

where ˆ XYσ  is the sample covariance between the two variables (X and Y) and ˆiσ  is the standard deviation 
of the ith variable. 

 
This coefficient ranges from -1.0 to +1.0, and takes the value of zero when no linear relationship between X and Y is 
present in the sample. The Pearson Product-Moment Correlation Coefficient has proven useful as a simple summary 
statistic for describing bivariate relationships, but it also serves as an important building block for more advanced 
statistical analyses such as multiple regression analysis and factor analysis. 
 
Inferences about the direction and strength of relationships in populations, rather than samples, are frequently made 
using both null hypothesis testing (i.e., testing the null hypothesis that the population coefficient of correlation, ,XYρ  is 
zero) and confidence interval estimation. In both forms of inferential statistics, the sample correlation coefficient is 
used as a point estimate of the population coefficient, that is: 
 

ˆXY XYr ρ=  
 
A variety of research artifacts are known to weaken or attenuate the apparent correlation between variables, including 
artificial restriction of variability, nonlinearity, and measurement error in the variables. Adjustments for such 
attenuation are needed to reduce or remove the statistical bias in the coefficient. The SAS macro presented in this 
paper provides an easy method for disattenuating sample coefficients calculated from variables that are measured 
with random error. 

IMPACT OF MEASUREMENT ERROR ON CORRELATION 

Measurement error has been identified as an important factor in the attenuation of correlation for many years. The 
classic paper by Spearman (1904) brought attention to the phenomenon. True-score theory posits that the observed 
values of variables (Xi) are functions of the true values of these variables (Ti) and errors of measurement (Ei): 

i i iX T E= +  

Under the typical assumptions of true-score theory (that errors of measurement are uncorrelated with true scores and 
uncorrelated with errors of measurement in other variables), measurement error in variables leads to attenuation of 
the apparent correlation between the variables, leading to observed correlations that are systematically smaller than 
the correlations between the true scores. The amount of measurement error in variables is usually represented by an 
index of reliability: 

where 2
Tσ is the variance of the true scores and 2

Eσ is the variance of the measurement errors. 

The index of reliability is the proportion of observed score variance that is variance in true scores rather than 
measurement error. The index ranges from zero to one, with higher values indicating measures containing less 
measurement error. 

CORRECTION FOR ATTENUATION AND ESTIMATION OF CONFIDENCE INTERVALS 

Spearman’s (1904) correction for attenuation derives directly from true-score theory: 

where *

XYρ is the corrected, or disattenuated, correlation (that is, the correlation between the true scores), 

XYρ is the correlation between the observed scores, and the iiρ are the reliabilities of the two variables. 
 

Although Spearman’s correction for attenuation provides a reasonable point estimate for the correlation between two 
variables after adjusting for errors of measurement, the construction of confidence intervals around this correlation 
has been more difficult. The traditional approach to confidence interval estimation (e.g., Hunter & Schmidt, 1990; 
Winne & Belfrey, 1982) uses the Fisher (1950) normalization of the obtained correlation coefficient to construct a 
confidence interval around the uncorrected coefficient, then disattenuates the endpoints of the resulting interval. 
Thus, for a 95% confidence interval around an observed sample correlation ˆXYρ : 

The endpoints of this interval are then back-transformed to the correlation metric: 

In the final step, the endpoints are disattenuated using Spearman’s (1904) correction for attenuation. 

 

An alternative approach to interval estimation was suggested by Charles (2005). Using extensive simulations, Charles 

derived upper and lower limits to the intervals as: 

MACRO ATTENUATE 

A SAS macro was designed to compute the disattenuated point estimate of the correlation, and the endpoints of both 
methods of interval estimation. The macro was developed to provide researchers with an easily accessible tool for 
calculating these statistics. Arguments supplied to the macro include the name of the SAS dataset that contains the 
other arguments, the uncorrected sample correlation coefficient, the sample size, reliability estimates for both 
variables, and the desired level of confidence for the intervals. In addition, an indicator argument is used to request 
html output via the SAS Output Delivery System (ODS). By default, the macro uses the latest SAS data set created, 
variable reliability values of .80, a 95% level of confidence, and no ODS output. 
 
The macro initializes a new SAS data step and performs the relevant computations for the disattenuation and the 
interval estimation. The output from the macro includes a table to present the calculated statistics, as well as 
descriptive information about the data analyzed. Of course, the macro syntax may be easily modified to write the 
output to a disk file or to send the statistics to regular SAS for further analyses. 
 
ATTENUATE Code 

%macro ATTENUATE (DATASET = _LAST_, N = N, OBS_CORR = rxy, RXX_1 = rxx1, RXX_2 = rxx2,  
Confidence = .95, WantODS = 0); 

 
* +-------------------------------------------------------------------------------------------+ 
 Macro Arguments: 
  DATASET = SAS Dataset containing summary data on the correlation 
  N = Sample size on which correlation was computed 
  OBS_CORR = Observed sample Pearson Product-Moment Correlation Coefficient 
  RXX_1 = Estimated reliability for variable 1 
  RXX_2 = Estimated reliability for variable 2 
  Confidence = Level of confidence for interval estimates 
  WantODS = Indicator argument to request ODS output as html (set WantODS = 1) 
  +-------------------------------------------------------------------------------------------+; 
 
 * +---------------------------------------------+ 
  Enable html Output Through ODS if Requested 
   +---------------------------------------------+; 
  %if &wantODS = 1 %then %do; 
   OPTIONS FORMCHAR="|----|+|---+=|-/\<>*"; 

   ods html;  
   ods html body='your-html-file.html'; 
  %end; 
 
data temp; 
 set &dataset; 
 CL = &confidence * 100; 
 
  * +-------------------------------------------------+ 
      Disattenuate the Sample Correlation Coefficient 
    +-------------------------------------------------+; 
 dis_corr = &obs_corr/(sqrt(&rxx_1)*sqrt(&rxx_2)); 
 
  * +------------------------------------------------------------+ 
   Compute Mean Reliability Estimate for the Two Variables 
    +------------------------------------------------------------+; 
 mn_reliability = (&rxx_1 + &rxx_2) / 2; 
 
 * +----------------------------------------------------------------------------------------+ 
  Compute Endpoints of Fisher CI Around Observed Correlation (without disattenuation) 
   +----------------------------------------------------------------------------------------+; 
    critZ = -1*probit((1-&Confidence)/2);    
    transfish_z = 0.5*log((1+abs(&obs_corr))/(1-abs(&obs_corr))); 
    if &obs_corr < 0 then transfish_z = transfish_z * -1;  
 SE_fish_z = sqrt((&N - 3)**-1); 
 upperCI = (exp(2*(transfish_z  + critZ*SE_fish_z)) - 1) / 

(exp(2*(transfish_z  + critZ*SE_fish_z)) + 1); 
 lowerCI = (exp(2*(transfish_z  - critZ*SE_fish_z)) - 1) / 

(exp(2*(transfish_z  - critZ*SE_fish_z)) + 1); 
 
 * +------------------------------------------------------+ 
  Disattenuate the Endpoints of the Fisher Interval 
   +------------------------------------------------------+; 
 dis_upper = UpperCI/(sqrt(&rxx_1)*sqrt(&rxx_2)); 
 dis_lower = LowerCI/(sqrt(&rxx_1)*sqrt(&rxx_2)); 
 
 * +----------------------------------------------------------------------+ 
  Compute the endpoints of the confidence sets suggested by Charles 
   +----------------------------------------------------------------------+; 
 L = -.0608*(1/SQRT(&N)) - .00455*(1/mn_reliability) + 1.111*(1/SQRT(&N)*1/mn_reliability); 

M = .00759 - 1.12*(1/SQRT(&N)) - .00355*(1/mn_reliability) +  
.152*(1/SQRT(&N)*1/mn_reliability); 

 
 Charles_low = (-1 + SQRT(1 - 4*(critZ*M)*(critZ*L - dis_corr)))/(2*(critZ*M));   
 Charles_high = (-1 + SQRT(1 - 4*(-critZ*M)*(-critZ*L - dis_corr)))/(2*(-critZ*M));  
 
 * +------------------------------------------+ 
  Create Macro Output Using File Print 
   +------------------------------------------+; 
 file print header = h notitles; 
 put @5 CL 2. @7 '%' @17 LowerCI 6.3 @27 UpperCI 6.3 @40 dis_lower 6.3 @50 dis_upper 6.3  
  @63 Charles_low 6.3 @74 Charles_high 6.3; 
 return; 
 
 h: put 
 @1 'Disattenuation of Correlation Coefficients' /// 
 @1 'Observed Correlation:' @35 &obs_Corr 6.3 // 
 @1 'Reliability Estimates:' @35 &rxx_1 6.3 / 
        @35 &rxx_2 6.3 // 
 @1 'Sample Size:'   @35 &N 6.0 // 
 @1 'Disattenuated Correlation:' @35 dis_corr 6.3 //// 

 @1 'Interval Estimates:' // 
 @1 '               Observed Correlation            Disattenuated Correlation' / 
 @1 '               --------------------     ---------------------------------------' / 
 @1 '                 Fisher Interval        Fisher Interval        Charles Interval' / 
 @1 'Confidence       ---------------        ---------------        ----------------' / 
 @1 '  Level          Lower     Upper        Lower     Upper        Lower      Upper' / 
 @1 '----------       -----     -----        -----     -----        -----      -----'; 
run; 
 * +---------------------------------------------+ 
  Close html Output File if it was Opened 
   +---------------------------------------------+; 
  %if &wantODS = 1 %then %do; 
   ods html close; 
  %end; 
* +--------------------+ 
   Delete TEMP Dataset 
  +--------------------+; 
 proc datasets; 
   delete temp; 
 run; 
 
%mend ATTENUATE; 
 
EXAMPLES OF MACRO ATTENUATE 

The easiest way in which the macro ATTENUATE may be used is to simply create a SAS dataset that inputs the 
sample summary data. The macro is then called, using as arguments the name of the dataset and the names of the 
relevant variables. Summary data from 100 observations are used to illustrate the macro. These data provided a 
sample correlation of .70, and the reliability estimates for the two variables were .80 and .90. The observed data are 
read into the SAS dataset ONE. The call to the macro requests confidence intervals of 90%, 95%, and 99%. For this 
example, no html output has been requested, so the macro output is written to the output window in SAS. 
 
data one; 
 input sampsize r reliable1 reliable2 conf;    
  cards; 
 100 .70 .80 .90 .90 
 100 .70 .80 .90 .95 
 100 .70 .80 .90 .99 
 ; 
 
Title 'Disattenuation of Correlation Coefficients: No HTML Ouput'; 
 %attenuate (dataset = one, n = sampsize, obs_corr = r, rxx_1 = reliable1, rxx_2 = reliable2,  
  confidence = conf); 
 
As a second example, summary data from 200 observations provided a sample correlation of .62, with reliability 
estimates for the two variables of .70 and .80. The observed data are read into the SAS dataset TWO. As with the 
first example, the call to the macro requests confidence intervals of 90%, 95%, and 99%. For this example, html 
output has been requested (the WantODS = 1 argument in the macro call). 
 
data two; 
 input sampsize r reliable1 reliable2 conf;    
  cards; 
 200 .62 .70 .80 .90 
 200 .62 .70 .80 .95 
 200 .62 .70 .80 .99 
 ; 
Title 'Disattenuation of Correlation Coefficients: Requested HTML Ouput'; 
 %attenuate (dataset = two, n = sampsize, obs_corr = r, rxx_1 = reliable1, rxx_2 = reliable2,  
  confidence = conf, WantODS = 1); 
run; 

OUTPUT FROM MACRO ATTENUATE 

Table 1 provides the macro output from the first example. The output displays the values of the arguments sent when 
the macro was invoked (the observed correlation, reliability estimates, and the sample size). The point estimate of the 
disattenuated correlation coefficient (.825) is provided, followed by interval estimates based on the observed (e.g., 
uncorrected) correlation coefficient, and interval estimates of the disattenuated correlation using the Fisher/Spearman 
and the Charles approaches.  The macro program allows for the researcher to garner a point estimate of the 
disattenuated correlation coefficient for any observed value, reliability estimate and sample size.   
 
Although the observed sample correlation is only .70, after adjusting for the measurement error in the two variables 
(i.e., rii = .80 and .90 for the two variables) the expected correlation is notably stronger (r = .825). Without adjusting for 
the attenuation of the correlation, the 95% confidence interval (the second row of the interval estimates part of the 
printed output) ranges from .584 to .788. In contrast, following the disattenuation the Fisher/Spearman 95% 
confidence interval ranges from .688 to .929. The Charles approach yields an interval that is very similar to the 
Fisher/Spearman interval: with the Charles approach, we can say we are 95% confident that the population 
correlation is between .671 and .911 if the two variables were measured with perfect reliability. 
 
Table 1 
Sample Output from Macro. 

 
 
Disattenuation of Correlation Coefficients 
 
 
Observed Correlation:              0.700 
 
Reliability Estimates:             0.800 
                                   0.900 
 
Sample Size:                         100 
 
Disattenuated Correlation:         0.825 
 
 
 
Interval Estimates: 
 
               Observed Correlation            Disattenuated Correlation 
               --------------------     --------------------------------------- 
                 Fisher Interval        Fisher Interval        Charles Interval 
Confidence       ---------------        ---------------        ---------------- 
  Level          Lower     Upper        Lower     Upper        Lower      Upper 
----------       -----     -----        -----     -----        -----      ----- 
    90%          0.605     0.776        0.712     0.914        0.702      0.900 
    95%          0.584     0.788        0.688     0.929        0.671      0.911 
    99%          0.541     0.811        0.638     0.955        0.603      0.930 

 
 
Table 2 provides the html macro output from the second example. The output displays the values of the arguments 
sent when the macro was invoked (the observed correlation, reliability estimates, and the sample size). The point 
estimate of the disattenuated correlation coefficient (.829) is provided, followed by interval estimates based on the 
observed (e.g., uncorrected) correlation coefficient, and interval estimates of the disattenuated correlation using the 
Fisher/Spearman and the Charles approaches.  The interval estimates of the dissattenuated correlation coefficient 
using the Fisher/Spearman and the Charles approaches produce confidence bands that are slightly wider than those 
around the obtained correlation coefficient and, of course, suggest the correlation between the variables without 
measurement error will be stronger than that observed.   
 
Although the observed sample correlation is only .620, after adjusting for the measurement error in the two variables 
(i.e., rii = .70 and .80, respectively) the expected correlation is notably stronger (r = .829). Without adjusting for the 
attenuation of the correlation, the 95% confidence interval (the second row of the interval estimates part of the printed 

output) ranges from .494 to .720 In contrast, following the disattenuation the Fisher/Spearman 95% confidence 
interval ranges from .660 to .953. The Charles approach yields an interval that is very similar to the Fisher/Spearman 
interval: with the Charles approach, we can say we are 95% confident that the population correlation is between .654 
to .933 if the two variables were measured with perfect reliability. 
 
 
SIMULATION RESEARCH ON CONFIDENCE INTERVALS 

Kromrey, Fay, and Bellara (2008) conducted a simulation study to investigate the accuracy and precision of the 
confidence intervals for disattenuated correlation coefficients.  In this Monte Carlo study, random samples were 
generated under known and controlled population conditions. Samples were generated from bivariate, normal 
populations, measurement error was induced in each variable, and the correlation was estimated based on the 
sample coefficient and the disattenuated coefficient (with disattenuation based on both the known reliability (i.e., the 
true reliability based on the simulation  parameters) and the sample estimate of the reliability).  The Monte Carlo study 
included three factors in the design: (a) the true correlation in the population (with XYρ = .10, .30, .50, .70, and .90), 

(b) the reliabilities of the two variables (with XXρ = .60, .70, .80, and .90), and (c) sample size (with N = 25, 50, 100, 
250, 500, and 1000). 
 
For each condition investigated in the study, 10,000 samples were generated. The use of 10,000 estimates provides 
adequate precision for the investigation of the sampling behavior of point and interval estimates of the correlation 
coefficient. For example, 10,000 samples per condition provide a maximum 95% confidence interval width around an 
observed proportion that is ± .0098 (Robey & Barcikowski, 1992). The outcomes of interest in this simulation study 
included both point estimates (the bias and sampling error of the observed and disattenuated correlation  coefficients) 
and interval estimates (confidence interval coverage and width for interval estimates based on the observed 
coefficient, the Fisher/Spearman interval, and the Charles (2005) interval). 
 
An overall summary of the confidence interval results is provided in Figures 1 – 4. For each condition investigated, the 
proportion of confidence intervals that contained the population parameter from the data generation was calculated. 
These proportions provide estimates of the interval coverage probabilities. Figure 1 provides a distribution of these 
coverage probabilities across all conditions investigated in the simulation study, when the disattenuation and the 
interval estimates were based on the known reliabilities of the variables. The figure presents the distributions of the 
coverage probabilities for confidence intervals constructed around the observed correlation (no disattenuation), and 
those constructed using the Fisher/Spearman approach and the Charles approach. Note the very poor coverage in 
many conditions for confidence intervals constructed without disattenuation. In contrast, both the Fisher/Spearman 
and the Charles intervals provide near nominal-level coverage (95%) across most conditions. The Fisher/Spearman 
interval coverage was slightly more consistent in coverage probabilities than the Charles intervals. 
 
Figure 2 also illustrates the distribution of interval coverage probabilities across all conditions investigated in this 
study, but compares the disattenuation and interval estimates based on the sample reliabilities. These data, like those 
presented in Figure1, also reveal the distributions of coverage probabilities for both the observed correlation 
coefficients and the Fisher/Spearman and Charles intervals for the disattenuated correlation coefficients. The 
intervals for the observed coefficients still show poor coverage compared to the other two.  However, in this case, the 
variability in the coverage probabilities between the Charles and Fisher/Spearman methods appears to be negligible.  
 
The final two figures display the distributions of the confidence interval widths for the observed coefficients (without 
disattenuating), for the disattenuated coefficients using Fisher/Spearman and for the disattenuated coefficients using 
the Charles method. Figure 3 shows the distribution of the widths using the known reliabilities, while Figure 4 displays 
the bands computed using sample reliability estimates. It can be seen in both figures (in both sets of intervals) that 
the differences in confidence interval widths between the Fisher/Spearman and Charles methods once again appear 
to be negligible. It is, however, patently obvious that the concomitant confidence interval bands for the observed 
correlation coefficients are much narrower in width than the two disattenuated intervals.  

Table 2 
Sample HTML Output from Macro. 
 
 
 

Disattenuation of Correlation Coefficients                                              
                                                                                         
                                                                                        
Observed Correlation:              0.620                                                 
                                                                                         
Reliability Estimates:             0.700                                                 
                                   0.800                                                 
                                                                                        
Sample Size:                         200                                                 
                                                                                         
Disattenuated Correlation:         0.829                                                 
                                                                                         
                                                                                        
                                                                                         
Interval Estimates:                                                                     
                                                                                         
               Observed Correlation            Disattenuated Correlation                 
               --------------------     ---------------------------------------          
                 Fisher Interval        Fisher Interval        Charles Interval          
Confidence       ---------------        ---------------        ----------------          
  Level          Lower     Upper        Lower     Upper        Lower      Upper          
----------       -----     -----        -----     -----        -----      -----          
    90%          0.543     0.687        0.725     0.918        0.727      0.901          
    95%          0.527     0.699        0.704     0.934        0.704      0.912          
    99%          0.494     0.720        0.660     0.963        0.654      0.933          

Figure 1. Distribution of Estimated Confidence Interval Coverage Using True Reliability 

Figure 2. Distribution of Estimated Confidence Interval Coverage Using Sample Reliability 

Figure 3. Distribution of Estimated Confidence Interval Width Using True Reliability 

Figure 4. Distribution of Estimated Confidence Interval Width Using Sample Reliability 

CONCLUSION 

The potential to adjust correlation coefficients for attenuation resulting from measurement error is an important tool for 
researchers to consider. Empirical research has demonstrated the importance of being mindful with regard to 
measurement error in the interpretation of research results. That is, point estimates and interval estimates based on 
the observed correlation coefficient become notably less useful (with increasing statistical bias and decreasing 
accuracy) as the degree of measurement error in the observed variables increases. Of course, the presentation and 
elucidation of disattenuated correlation coefficients requires the concomitant articulation of appropriate caveats (e.g., 
to avoid being accused of ‘interpreting what the data would have yielded if they had been psychometrically better than 
they really are!’). 
 
The macro ATTENUATE is provided to facilitate researchers’ calculation and use of both point and interval estimates 
of disattenuated correlation coefficients. The macro is easy to use and is written entirely in Base SAS software, with 
all calculations and output provided through a SAS data step. Further, the macro may be easily modified to retain the 
disattenuated point estimates and the confidence interval endpoints in a SAS data set for use in further analysis or for 
graphical display. 

